

第7章 股票市场风险指标计算

清华大学经管学院 朱世武

Zhushw@sem.tsinghua.edu.cn

Resdat样本数据: www.resset.cn

SAS论坛: www.resset.cn

计算指标与环境

理论模型 $r_j = \alpha_j + \beta_j r_m + \varepsilon_j$

其中, r_j 为股票j的超额收益; r_m 为市场超额收益; α_j 和 β_j 为参数, 且假定 β_j 和 ε_j 不相关。

将方差作为风险的度量, 根据上述假设得到股票j的总风险为:

$$\sigma_j^2 = \beta_j^2 \sigma_m^2 + \sigma_{\varepsilon_j}^2$$

选定时期, 根据经验数据可以得模型参数的最小二乘估计 $\hat{\alpha}_j, \hat{\beta}_j$ 以及相应估计值 $\hat{\sigma}_j^2, \hat{\sigma}_m^2$ 和 $\hat{\sigma}_{\varepsilon_j}^2$ 。于是, 系统风险占该股票总风险的比例估计值为:

$$\frac{\hat{\beta}_j^2 \hat{\sigma}_m^2}{\hat{\sigma}_j^2} = \hat{R}_{jm}^2 \quad \text{有} \quad \hat{R}_{jm}^2 = \frac{\sum_{i=1}^n (\hat{Y}_i - \bar{Y})^2}{\sum_{i=1}^n (Y_i - \bar{Y})^2} = \frac{\hat{\beta}_j^2 \sum_{i=1}^n (X_i - \bar{X})^2}{\sum_{i=1}^n (Y_i - \bar{Y})^2}$$

The Power to Know.

计算指标

计算1999至2005年，沪深两地股票市场系统风险系数、个股总风险和个系统风险占总风险比例三个风险指标等。即具体计算：

系统风险系数估计值 $\hat{\beta}$ ；

个股总风险估计值 $\hat{\sigma}$ ；

系统风险占总风险比估计值 \hat{R}^2 ；

全市股票上述三种指标的概况统计量。

计算环境

计算数据集：

银行存款利率Banklr;

基准利率Bchmklr;

个股日收益Dret;

市场日收益DretM。

时期：2005年。

样本：沪深两市场满足条件的2006年前上市所有股票日百分比收益。剔除1年内交易小于50天的股票。

算法实现

第一步：创建无风险利率数据集RF；2005年市场日收益数据。

第二步：创建宏文本：Stk2006.TXT。

第三步：计算2005年沪深两市场A股个股风险指标。

第四步：沪深两市场个股与市场风险关系统计分析。

计算程序

创建无风险利率数据集:

/*1998年7月1日后使用七日回购利率两周指数加权平均为基准利率 */

data a;

set ResDat.bchmkir;

if code="B2W" and date>='1jun1998'd;

rename ir=rf;

keep ir date;run;

/*1998年7月1日前用一年期银行存款利率加10%为基准利率*/

data b;

set ResDat.bankir;

if code="d1y" ;

run;

data c;

set b;

format date yymmdd10.;

if enddt=. Then enddt=date();

do date=begdt to enddt;

output;

end;

run;

The Power to Know.

```
data d;  
set c;  
if date<'1jun1998'd;  
keep date rf;  
rf=ir*1.1; /* 在一年期存款利率的基础上再增  
加10% */  
run;  
/*合并得到基准利率表rfyr*/  
data rfyr;  
set d a;  
if date<='31dec2005'd;  
run;  
/*得到日基准利率表rf*/  
data rf;  
set rfyr;  
rf=rf/365; /* 将利率转化为日利率 */  
run;
```

2005年市场日收益数据DretM2005：总市值加权平均市场日收益率。

```
Data DretM2005;  
Set ResDat.DretM;  
If year(date)=2005 and Exchflg='0' and Mktflg='AB';  
/* Exchflg='0'—沪、深两交易所；Mktflg='AB'—AB股票市场  
*/  
Keep date Dretmc;  
Run;
```


创建宏文本：Stk2006.TXT。

利用选择的样本股票收益数据集Dret创建宏。如果要计算全部股票的风险指标，可以直接利用lstkinf创建宏。

```
Data a;  
Set ResDat.dret;  
If dif(stkcd)=0 then delete;  
data _null_;  
set a;  
a='%a(';   
b=',';  
c=')';  
file "Stk2006.txt" ;  
put a $ stkcd $ b $ lstknm $ c $ ;  
run;
```

```
%a( 000001 , S深发展A );  
%a( 000002 , 万科A );  
%a( 000004 , *ST国农 );  
%a( 000005 , ST星源 );  
%a( 000007 , 深达声A );  
%a( 000009 , S深宝安A );  
%a( 000011 , S深物业A );  
%a( 000012 , 南玻A );  
%a( 000016 , 深康佳A );  
%a( 000017 , S*ST中华 );  
%a( 200011 , 深物业B );  
%a( 200012 , 南玻B );  
.....
```

计算2005年沪深两市场个股风险指标：模型 $r_s = a + \beta r_m + e$
/* 2005年无风险利率数据集rf2005 */
Data rf2005;
Set rf;
If year(date)=2005;
Run;
/* 2005市场超额收益数据集 */
Data PDretM2005;
Merge DretM2005 rf2005;
By date;
rm= Dretmc- rf;
keep date rm;
run;
data PDretM2005; /* 收益数据缺失时，用前面交易日的数据填充 */
set PDretM2005;
retain r_m;
if rm^=. Then r_m=rm;
drop rm;
run;
/* 2005年个股票超额收益数据集 */
Data dret;
Set ResDat.Dret;
Where year(date)=2005;
Run;

The Power to Know.

```
Proc sql;
Create table Pdret2005 as
Select * from Dret left join rf
On Dret.date=rf.date;
Run;
Proc sort data= Pdret2005;
By stkcd date;
Run;
Data Pdret2005;
Set Pdret2005;
r_s=dret-rf;
keep stkcd lstknm date r_s;
run;
Proc sql;
Create table Capm2005 as
Select * from Pdret2005 left join PdretM2005
On Pdret2005.date=PdretM2005.date;
Run;
Proc sort data=capm2005;
By stkcd;
Run;
```

The Power to Know.

```


/* 计算个股风险指标：数据集Capm2005 */
proc reg data=capm2005 outest=d cp noprint; /*输出参数估计
数据集中包括Mallow Cp估计量 */
model r_s=r_m; /* 回归模型为： $r_s=a+\beta r_m+e$  */
by stkcd;
run;
data d;
set d;
if (_EDF+_P_)<50 then delete; /* 删除2005年交易日小于50天
的股票，_EDF_为残差自由度，_P_为参数个数 */
data RiskI2005 (keep=sn obs _rmse_ r_m _rsq_ stkcd
label="2005年个股与市场风险指标" );
set d;
sn=_n_;
obs=(_edf+_p_);
label sn='个股序号';
label obs='观测个数';
label _rmse_='个股风险度';
label r_m='系统风险估计值beta';
label _rsq_='系统风险占总风险比例';
run;

```


The Power to Know.

对数据集Riskl2005中个股的三种风险指标作图：

```
%macro a(x);  
proc gplot data= Riskl2005;  
plot &x*sn=1 ;  
symbol1 v=star i=none r=1 c=blue;  
proc plot data=rskindx2005 vpct=80;  
plot &x*sn ;  
%mend a;  
%a(r_m);  
%a(_rmse_);  
%a(_rsq_);  
run;
```

系统风险占总风险比例



2005年股票市场个股风险指标的概括统计量计算：

```
proc univariate data= RiskI2005 noprint;
var r_m _rmse_ _rsq_ ;
output out=out n=n_r_m n_rmse_ n_rsqu_
mean=mean_r_m mean_rmse_ mean_rsqu_
median=median_r_m median_rmse_ median_rsqu_
min=min_r_m min_rmse_ min_rsqu_
max=max_r_m max_rmse_ max_rsqu_
q1=q1_r_m q1_rmse_ q1_rsqu_
q3=q3_r_m q3_rmse_ q3_rsqu_
range=range_r_m range_rmse_ range_rsqu_;
data out;
set out;
q3_q1_r_m =q3_r_m -q1_r_m ;
q3_q1__rmse_ = q3_rmse_-q1_rmse_ ;
q3_q1_rsqu_ = q3_rsqu_-q1_rsqu_ ;
format _numeric_ 6.4;
run;
```

The Power to Know.